
%let seed_pt = %eval(&seed*10) ;

/*proc import file = "&par_path/parameters_BE_model2.csv" out = 
par_be dbms=csv replace ;
run ;*/

data par_pt ;
infile "&par_path/parameters_PT_model2.csv" DSD MISSOVER delimiter = 
"," termstr = CRLF lrecl = 32767 firstobs = 2 ;
informat variable $15. ;
informat estimate BEST12. ;
informat stderr BEST12. ;
informat tvalue BEST12. ;
informat probt $10. ;
format variable $15. ;
format estimate BEST12. ;
format stderr BEST12. ;
format tvalue BEST12. ;
format probt $10. ;
input variable $
      estimate
      stderr
      tvalue
      probt $ ;
run ;

/*
proc import file = "&par_path/vres_BE_model2.csv" out = v_be 
dbms=csv replace ;
run ;*/

data v_pt ;
infile "&par_path/vres_PT_model2.csv" DSD MISSOVER delimiter = "," 
termstr = CRLF lrecl = 32767 firstobs = 2 ;
informat v BEST12. ;
format v BEST12. ;
input v ;
run ;

data _null_ ;
set v_pt ;
call symput("sd",sqrt(v)) ;
run ;

/* generate u following a uniform law */

proc sort data = h_pt ;
by sa0100 sa0010 im0100 ;
run ;

data h_pt ;
set h_pt ;
retain x1 &seed_pt x2 0 ;
if _n_ = 1 then do ;
    x1 = &seed_pt ;
    x2 = 0 ;
end ;



else do ;
    y = mod(x1*20077+12345,65536) ;
    x2 = mod(int((x1*20077+12345-y)/65536)+mod(16838*x1+20077
*x2,65536),32768) ;
    x1 = y ;
end ;
z = 65536*x2+x1 ;
u = z/2147483648 ;
drop y z x1 x2 ;
run ;

/* compute the consumption */

data par_pt ;
set par_pt ;
call symput(compress("par"!!_n_),estimate) ;
run ;

data h_pt ;
set h_pt ;
/* reshape covariates */
cfood = hi0100*12 ;
cresto = hi0200*12 ;
rent = max(hb2300*12,0) ;
l_cfood = log(max(cfood,1)) ;
l_cresto = log(max(cresto,1)) ;
l_rent = log(max(rent,1)) ;
head_male = (ra0200 = 1) ;
tenure_2 = (hb0300 = 3) ;
tenure_3 = (hb0300 = 4) ;
hhsize_1 = (dh0001 = 1) ;
hhsize_3 = (dh0001 >= 3) ;
agerp_1 = (ra0300 < 30) ;
agerp_2 = (30 <= ra0300 < 40) ;
agerp_3 = (40 <= ra0300 < 50) ;
agerp_4 = (50 <= ra0300 < 60) ;
agerp_5 = (60 <= ra0300 < 70) ;
agerp_6 = (ra0300 >= 70) ;
number_children_1 = (number_children = 1) ;
number_children_2 = (number_children = 2) ;
number_children_3 = (number_children = 3) ;
labour_status_1 = (pe0100a in (1,2)) ;
labour_status_2 = (pe0100a in (3,4,6,7,8,9)) ;
labour_status_3 = (pe0100a = 5) ;
diploma_1 = (pa0200 = 1) ;
diploma_2 = (pa0200 = 2) ;
diploma_5 = (pa0200 = 5) ;
run ;

proc sort data = h_pt ;
by im0100 ;
run ;



proc univariate data = h_pt ;
by im0100 ;
var di2000 ;
weight hw0010 ;
output out = perc_implicates pctlpts = 20 40 60 80 pctlpre=p ;
run ;

proc univariate data = perc_implicates ;
var p20 p40 p60 p80 ;
output out = perc mean = p20 p40 p60 p80 ;
run ;

data perc;
set perc ;
call symput("quint1",p20) ;
call symput("quint2",p40) ;
call symput("quint3",p60) ;
call symput("quint4",p80) ;
run ;

data h_pt ;
set h_pt ;
income_quintile_1 = (di2000 <= &quint1) ;
income_quintile_2 = (&quint1 < di2000 <= &quint2) ;
income_quintile_3 = (&quint2 < di2000 <= &quint3) ;
income_quintile_4 = (&quint3 < di2000 <= &quint4) ;
income_quintile_5 = (di2000 > &quint4) ;
run ;

/* compute consumption */

data h_pt ;
set h_pt ;
/* bounds */
lbound = cfood+cresto+rent ;
a = log(lbound) ;
Xbeta = &par1+&par2*l_cfood+&par3*(l_cfood**2)+&par4*(l_cfood**3)
    +&par5*income_quintile_2+&par6*income_quintile_3+&par7
*income_quintile_4+&par8*income_quintile_5
    +&par9*l_cfood*income_quintile_2+&par10*(l_cfood**2)
*income_quintile_2+&par11*(l_cfood**3)*income_quintile_2
    +&par12*l_cfood*income_quintile_3+&par13*(l_cfood**2)
*income_quintile_3+&par14*(l_cfood**3)*income_quintile_3
    +&par15*l_cfood*income_quintile_4+&par16*(l_cfood**2)
*income_quintile_4+&par17*(l_cfood**3)*income_quintile_4
    +&par18*l_cfood*income_quintile_5+&par19*(l_cfood**2)
*income_quintile_5+&par20*(l_cfood**3)*income_quintile_5
    +&par21*l_cresto+&par22*(l_cresto**2)+&par23*(l_cresto**3)
    +&par24*l_rent+&par25*(l_rent**2)+&par26*(l_rent**3)
    +&par27*agerp_2+&par28*agerp_3+&par29*agerp_4+&par30*agerp_5
+&par31*agerp_6
    +&par32*head_male
    +&par33*tenure_2+&par34*tenure_3
    +&par35*hhsize_1+&par36*hhsize_3



    +&par37*number_children_1+&par38*number_children_2+&par39
*number_children_3
    +&par40*labour_status_2+&par41*labour_status_3 ;
Phi_a = cdf('NORMAL',a-Xbeta,0,&sd) ;
DI3001 = round(exp(Xbeta+quantile('NORMAL',Phi_a + (1-Phi_a)*u,0,
&sd))) ;
run ;


